
ABSTRACT  
Analysts, in particular SAS programmers, are often tasked with compiling data and generating reports at 
a moment’s notice. It is paramount to make report generation efficient for ease of replication. The method 
of preparing and compiling recurring reports can be made easier when the process of reporting is well-
defined. The objectives for this paper are to outline a checklist that can guide the creation of a reporting 
process for your data. SAS capabilities like labels and formats as well as the SAS Macro Facility play an 
important role in automating a reporting process. Identifying SAS procedures ahead of time, like PROC 
PRINT or PROC TABULATE, and developing a program that is flexible and adaptable can save time. 
General steps to successfully develop a reporting process will be described and detailed examples to 
show how to achieve this will be provided. The overall benefits of planning a reporting process are to 
reduce reporting errors, reduce programming time and have a well-designed report that is meaningful.   

INTRODUCTION  
There are four important points laid out in this paper that will guide you in the process of developing a 
SAS program that can efficiently and effectively generate reports on a recurring basis. These are general 
concepts and their applications range greatly depending on the reporting process you wish to achieve as 
well as the SAS platform you may be using. Examples are shared for each of these points to provide 
more insight and explanation. 

 

Plan your report 

Rely on data sources that are stable and reliable 

Organize your reporting process by program function 

Consider using SAS macros to reduce your code 

 

SAS Enterprise Guide (EG) was used to develop the example report for this paper, but Base SAS can be 
used as well. The final report generated is in a PDF format using the SAS Output Delivery System (ODS) 
and can be viewed on the following page. 

THE FINAL PDF REPORT USED FOR THE EXAMPLES IN THIS PAPER 

PLAN YOUR REPORT 
It is a good idea to visualize how your final report will look before programming. The first step in 
developing a reporting program involves creating a mockup or draft of what your report will look like. If 
you intend to deliver the report in excel you can develop a template using that software. A less time 
consuming strategy is to draw it out the old fashioned way on paper. Below is an image of a primitive 
example of a report mock up. It may not be fancy, but it serves its purpose. 

 
 
ELEMENTS IN YOUR REPORT 
Taking the time to envision the layout of your report will 
aid in the programming process. It can help determine 
the SAS procedures you will require as well as identify 
places in your program that can be simplified with SAS 
macros and other programming logic. Pay particular 
attention to areas of your report that will require the use 
of SAS procedures (PROC TABULATE, PROC 
MEANS). If you find the tables you wish to display 
within your report are similarly structured, this is an 
opportunity to reduce your SAS code. We will come 
back to this idea later.    

 
 
USING ODS TO GENERATE THE FORMATTED FILE 
SAS has the ability to generate and produce reports in different formats using ODS capability. Although 
the ODS statement is used for much more than generating SAS output to other file formats, leveraging 
ODS statements in SAS can automate the process of converting SAS output to a format you intend to 
share with consumers of the report. There are several formats that ODS supports including Portable 
Document Format (PDF), Rich Text Format (RTF), HyperText Markup Language (HTML) and Comma 
Separated Values (CSV). Relatively new in SAS 9.4 are ODS statements for PowerPoint and Excel, 
although it should be noted that the ability to export output to an Excel file has been available for a while 
through other means like ODS ExcelXP Tagsets and Dynamic Data Exchange (DDE). The example in 
this paper relies on ODS PDF. To successfully produce a PDF document using ODS, simply begin and 
end your program with the ODS statements shown below.  

ods pdf; 

<SAS program> 

ods pdf close; 

 

Questions to ask yourself during this step…. 

• Which tables, charts or graphs will be included in the report? 

• Where do you want specific text or images to be placed (headers/footers, company logo, text/write-
up)?  

• How will you deliver your report? Which file format? (Excel, PDF) 

RELY ON DATA SOURCES THAT ARE STABLE AND RELIABLE  
The most sophisticated SAS program written to efficiently generate a recurring report will quickly break if 
the data is not available in a structure the program needs. In other words, a report cannot be generated 
quickly if the data is not stable and predictable. It is important to identify element names and table 
sources ahead of time and ensure these will be consistent for future reporting. If you are relying on data 
that are maintained by a third party, it is a good idea to communicate with them and assess the stability of 
the source. An additional thing to consider is when new data are available, so that your report is 
scheduled to run at a reasonable time. 

Questions to ask yourself during this step…. 

• Which sources of data do you need? (Tables, Elements) 

• How often are data refreshed and does the schedule for reporting account for this? 

ORGANIZE YOUR REPORTING PROCESS BY PROGRAM FUNCTION 
Organizing your program may differ depending on the SAS platform you are using. For example, if the 
report you are generating is lengthy and you are working in EG it may benefit you to separate program 
functions into separate nodes of code as has been done in Figure 1. 

  

Figure 1. EG Project used to create the report 

There are alternatives to structuring your SAS program, like utilizing the %include statement to call parts 
of your program from SAS files saved in other locations. You can always place your entire program into 
one program node, separating out sections with clear comments. If you have developed many SAS 
macro functions and you find you use them heavily across projects, a more advanced option is to create a 
permanent SAS Macro library by utilizing Compiled Stored Macros or the AUTOCALL options. Whatever 
your preference, the purpose is to clearly organize sections of your program so that editing or updating 
the program for any reason is easy and intuitive.  
 

Below describes the contents of each SAS program node from Figure 1. Many of the nodes of code listed 
contain a single SAS macro or define several for use in other parts of the program. 
 
Step 1 

• Labels: SAS macro that defines and houses all the labels for elements used in the report; This 
macro is later called in Step 2 - Get Data 

• Formats: SAS macro that defines and houses all the formats for elements used in the report; This 
macro is later called in Step 2 - Get Data 

• Footnotes: SAS macro that places any footnotes needed at the end of tables in the report; This 
macro is called in Table Functions 

• Table Functions: SAS program that houses several SAS macros used to generate tables in the 
report; These macros are called in Step 3 - Generate Report and mainly pass the element used to 
create the table 

• Global Variables: SAS program that defines SAS macro variables that are specific to the report 
content (year or date placed in the report, specific names of groups one is reporting on) 

• Report Format: SAS program that defines SAS macro variables that are specific to table or report 
formatting (PROC TEMPLATE, column widths for tables, background colors for table headers) 

 
Step 2 

• Get Data: Data processing required to structure the data for reporting (libname statements, data 
steps, PROC SQL) 

 
Step 3 

• Generate Report: Main SAS program to generate report; Uses ODS PDF statement and places 
text and tables as needed 

 

Examples of how some of these SAS programs interact is explained in the next section. 

CONSIDER USING SAS MACROS TO REDUCE YOUR CODE 
SAS Macros are a useful way to simplify any SAS program. As was previously mentioned, much of the 
SAS code written throughout the example from Figure 1 uses the SAS Macro Facility Language. 

 
LABELS AND FORMATS 
The example below shows how labels and formats are defined and applied to the SAS data set you are 
working with. In parts 1 and 2, the labels and formats are defined in separate macros and these macros 
are then called later in part 3 during data processing and preparation. There are, of course, alternative 
ways to do this. Labels and formats can be applied to data sets permanently or they can be defined 
during procedure statements. The decision of how to store and apply labels and formats is really a 
function of how your reporting process is organized and the nature of your data.  

 

Code found in “Labels” in EG Project 

 
%macro labelquestions; 
 
label 
 
ADVISING='Please choose up to 3 sources from which you received the MOST 
BENEFICIAL academic advising during your last 2 years at UCF (select all that 
apply)' 
ADVISING_1 = 'College advising office staff' 
ADVISING_2 = 'Athletic Advising (ASSA)' 

... <more SAS labels> 

; 
 
%mend; 

Code found in “Formats” in EG Project 

 
%macro response_format; 
 
proc format; 
 
picture pct  

0.005-100='009.9%'  
0-0.0049=' ' 
; 

 
value $ major 

'RESPONDENT_TOTAL_COUNT' = 'Number of Respondents' 
'MAJAVAIL_NEW' = 'In my major, the courses I needed were available' 
'OTRLARGE_NEW' = 'Outside my major, too many of my classes were too large' 

... 

; 

... <more SAS formats> 

 
%mend; 

 

1 

2 

Code found in “Get Data” in EG Project 

<multiple library statements to access SAS datasets> 
 
%response_format; 
 
data work.GSS_ALL; 

set gss1415.gssay_new   
    gss1516.gssay_new (drop=URE ) 
    gss1617.gssay_new (drop=timestamp) ; 
denom = 1; 
age=.; gender=.; race=. ; status=.; 
rename  
ACAD_PLAN = ACAD_PLAN_GSS finished = finished_GSS  
ACAD_YEAR_AWARDED = YEAR_GSS age_df = age CF_BOE_GENDER = gender 
CF_BOE_RACE = race CF_BOE_RCENT_TYPE = status ; 

run; 

... <more data manipulation> ... 

data work.GSS_PLAN; 
set work.GSS_ALL; 

%labelquestions; 
if YEAR_GSS = '2014-15' then year_num = 1; 
else if YEAR_GSS = '2015-16' then year_num = 2; 
else if YEAR_GSS = '2016-17' then year_num = 3; 
where acad_plan_GSS = "&plan"; 

run; 

... <more data manipulation> ... 

 
 
SIMPLIFYING TABLE STYLES 
By creating SAS macro variables and defining them in an easily accessible part of your program, you can 
make changes to your report quickly. Recall the final output of the PDF report example: 

 

Figure 2. Close up screenshot of tables from final PDF report 

 

The black and gold background colors for headers were defined as macro variables in part 4 and were 
then referenced later in all macros defined in the “Table Functions” program (part 5). Recall that there are 
several macros defined in “Table Functions,” many of which use PROC TABULATE to structure certain 
types of table output for the report. An example of one of these macros is shown below. By defining the 
header background colors as macro variables in one place in the program, you can make changes easily 
if ever the style needs to be altered.  

Code found in “Report Formats” in EG Project 

 
%let respondent_bandcolor = #ffea99;  
%let header_bandcolor = #404040; 
%let header_textcolor = white; 
 

Code found in “Table Functions” in EG Project 

 
%macro matrix_questions(varlist, var_group, q_stem); 
 
<additional SAS code> ... 
 
proc report data=Report spanrows style(header) =  
{background=&header_bandcolor foreground=&header_textcolor just=c vjust=m} 
style(column) = {background=white just=l vjust=m} 
style(report)={outputwidth=100%}; 
 
column _name_ ('Positive Endorsements+' ( _numeric_)) ; 
 
label &yr_labels; 
define _Name_ / "&q_stem"  style(header)={just=l} ; 
define YR_1 - YR_3  / style(column)={just=c Cellwidth=&stat_width_PR.%} 
style(header)={just=c }; 
 

compute _NAME_; 
if lowcase(_NAME_) = 'number of respondents' then do; 
call define(_row_,'style',"style={background=&respondent_bandcolor}"); 
end; 
endcomp; 

 
format _NAME_ $&var_group.. _numeric_ stacked.; 
run; 
 

...<additional SAS code> 
 

%mend; 

 
BRINGING IT ALL TOGETHER IN A MAIN PROGRAM  
I suggest having one part of your program serve as the main stage. This is where most, if not all, other 
parts (other SAS programs, macro functions, macro variables) are called for use. In the example from 
Figure 1, this is part 6 – the “Generate Report” program. Only a portion of the program from part 6 is 
shown as the code is quite lengthy. Any macro variables used throughout the text are highlighted and 
macro functions to produce tables are noted as well. 

 

Code found in “Generate Report” in EG Project 

 

/*paper options*/ 
Option orientation=portrait papersize=letter topmargin=0.3in 
bottommargin=0.3in leftmargin=0.5in rightmargin=0.5in nonumber nodate; 
ods escapechar='^'; 

4 

5 

/*open ODS PDF statement*/ 
ods layout start width=8.5in height=11in ; 
ods pdf style=Custom startpage=never notoc dpi=300; 
 
/*clear ALL titles and footnotes*/ 
title; footnote; 
 
/********************************************************************/ 
/*Cover Page*/ 
 
proc odstext; 
p '^S={preimage="\\TheTab_KGrgb_300ppi_85_75.png"'; 
p " "; 
p " "; 
p "^S={FONT_SIZE=18pt JUST=l FONT_FACE=arial}Student Perceptions Report for 
Academic Program Review^{newline}^{newline}"; 
p "^S={FONT_SIZE=12pt JUST=l FONT_FACE=arial}Program: 
&plan_name.^{newline}^{newline}"; 
p "^S={FONT_SIZE=12pt JUST=l FONT_FACE=arial}Class of &class_year"; 
p '^S={FONT_SIZE=12pt JUST=l FONT_FACE=arial COLOR=CXFFD700} 
_________________________________________________________________'; 
run; 
 

...<additional SAS code> ... 
 
ods pdf startpage=now; /*Page break*/ 
/********************************************************************/ 
 
/*add headers and footers to the remaining pages*/ 
title HEIGHT=9PT J=LEFT F=ARIAL bold 
'^S={preimage="\\TheTab_KGrgb_300ppi_85_75_80.png"  
posttext="^-200y   Student Perceptions Report for Academic Program Review || 
&plan_name. || &class_year."'; 
 
footnote1 HEIGHT=9PT J=LEFT F=ARIAL bold "Operational Excellence and 
Assessment Support" J=RIGHT F=ARIAL bold "&date"; 
 
footnote2 HEIGHT=9PT J=LEFT F=ARIAL bold "Division of Teaching and Learning" 
J=RIGHT F=ARIAL bold "Page ^{thispage}"; 
 
proc odstext; 
p "^S={FONT_SIZE=11pt JUST=l FONT_FACE=arial 
COLOR=&section_header_color}Experience in the Major^{newline}"; 
p "^S={JUST=l FONT_FACE=arial}The table below shows positive endorsements for 
survey questions related to student experiences in the 
major.^{newline}^{newline}"; 
p "Table 1"; 
run; 
 
 
%matrix_questions( 
varlist = 
MAJAVAIL_NEW MAJTIME_NEW MAJRANGE_NEW MAJF2F_NEW MAJWEB_NEW MAJLARGE_NEW 
PROFGOOD_NEW PROFAVAI_NEW MAJCOMP_NEW MAJTECH_NEW OTRGEP_NEW OTRTIME_NEW 
OTRAVAIL_NEW OTRLARGE_NEW, 
var_group = MAJOR,  

Insert table, macro defined 
in “Table Functions” 

q_stem = Experience in the Major* 
); 
 
ods pdf startpage=now; /*Page break*/ 
/********************************************************************/ 
 
proc odstext; 
p "^S={FONT_SIZE=11pt JUST=l FONT_FACE=arial 
COLOR=&section_header_color}Academic Advising Resources^{newline}"; 
p "The table below summarizes which sources students indicated were the most 
beneficial during their last 2 years at UCF."; 
p " "; 
p "Table 2"; 
run; 
 
%multioption_questions(ADVISING, 11);  
 
...<additional SAS code> ... 
 
/*close ODS layout and PDF statement*/ 
ods layout end; 
ods pdf close; 

 

Questions to ask yourself during this step…. 

• Is data processing/manipulation needed to get the data structured the way you need it? 

• How will you apply any labels or formats required for your tables and charts? 

• If you have identified certain SAS procedures to produce output, can these be written as macro 
functions? 

• Is there any text in the report that may need updating for future reports? If so, can these be written 
as a macro variables? 

 

CONCLUSION 
Developing a SAS program to generate recurring reports can be a somewhat time intensive process up 
front. However, the benefit is that producing reports is quick and updating your program for any reason is 
easy. Planning how your report will look beforehand can assist in the programming process like 
identifying SAS procedures required for the job and organizing parts of your program in an efficient way. It 
is important to remember that the source of your data plays a vital role in making the reporting program 
work for you. Last but not least, the SAS Macro Facility is a useful way to reduce code and simplify parts 
of your program. The next time you work on a project that will most likely be requested on an on-going 
basis, I encourage you to use the steps laid out in this paper and design a SAS reporting process of your 
own. 

  

Insert table, macro defined 
in “Table Functions” 

